

# НЕЯВНЫЙ МЕТОД РЕШЕНИЯ УРАВНЕНИЙ ЛАМИНАРНОГО ПОГРАНИЧНОГО СЛОЯ НА ПЛАСТИНЕ

Лабораторная работа №3, 4

## План занятия:


1. Цель работы
2. Описание решаемой задачи
3. Метод решения
4. Метод прогонки (метод Томаса)
5. Порядок проведения моделирования
6. Содержание отчёта
7. Контрольные вопросы

## 1. ЦЕЛЬ РАБОТЫ

Расчёт несжимаемого ламинарного динамического пограничного слоя на плоской пластине **по неявной схеме** с устранением нелинейности методом запаздывающих коэффициентов.

## 2. ОПИСАНИЕ РЕШАЕМОЙ ЗАДАЧИ

Целью сегодняшней лабораторной работы является расчёт несжимаемого ламинарного динамического пограничного слоя на пластине неявным методом. Схема течения представлена на рисунке и полностью соответствует задаче, решённой на предыдущей лабораторной работе.



По-прежнему будем считать плотность среды и динамическую вязкость постоянными ( $\rho = \text{const}$ ,  $\mu = \text{const}$ ), градиент давления вдоль пластины равен нулю и проекция вектора силы тяжести направлена поперёк течения. В итоге получим:

$$\begin{aligned} \frac{\partial w_x}{\partial x} + \frac{\partial w_y}{\partial y} &= 0, \\ w_x \frac{\partial w_x}{\partial x} + w_y \frac{\partial w_x}{\partial y} &= \nu \frac{\partial^2 w_x}{\partial y^2}. \end{aligned} \quad (1)$$

Здесь  $\nu = \mu/\rho$  - коэффициент кинематической вязкости.

Для решения данной системы уравнений необходимо поставить три граничных условия для искомых переменных  $w_x$ ,  $w_y$ . В рассматриваемом случае:

на входе:  $w_x(x, y) = w_x(0, y) = w_{x0}$ ;  $w_y(x, y) = w_y(0, y) = 0$ ;

на стенке:  $w_x(x, y) = w_x(x, 0) = 0$ ;  $w_y(x, y) = w_y(x, 0) = 0$ ;

во внешнем потоке:  $w_x(x, y) = w_x(x, \delta) = w_{x0}$ .


Впервые решение этой задачи было получено Блазиусом <sup>1</sup>. Из решения Блазиуса известно, что коэффициент сопротивления трения  $c_f/2 = \tau_w / \rho w_{x0}^2 = \mu (\partial w_x / \partial y)_w / \rho w_{x0}^2$  в зависимости от числа Рейнольдса, построенного по длине пластины  $Re_x = w_{x0} x / \nu$  равен:

$$\frac{c_f}{2} = \frac{0.332}{\sqrt{Re_x}}. \quad (2)$$

### 3. МЕТОД РЕШЕНИЯ

Численные методы решения любых дифференциальных уравнений сводятся к замене непрерывных функций и их производных на функции, определённые в конечном числе узлов расчётной сетки. Эта процедура, названная **дискретизацией** уравнений, приводит к тому, что дифференциальные уравнения можно представить в виде алгебраического уравнения (системы алгебраических уравнений), как правило, линейного и допускающего решение методами линейной алгебры. От выбора способа дискретизации производных в системе уравнений (1) во многом зависит точность, устойчивость и физичность получаемого решения. В данной лабораторной работе мы проведём дискретизацию уравнений **по неявной схеме** и получим их решение.

Как и раньше введём сеточные функции  $(w_x)_{i,j}$  и  $(w_y)_{i,j}$  для продольной и поперечной составляющей вектора скорости  $w$ . Сеточные функции определены в точках расчётной сетки  $(x_{i,j}, y_{i,j})$ . Примем, что узлы расчётной сетки расположены на равных расстояниях друг от друга (равномерная сетка) по оси  $x$  и по оси  $y$ , тогда кол-во узлов в расчётной сетке по оси  $x$ :  $n = L/\Delta x$ , а по оси  $y$ :  $k = \delta_L/\Delta y$ , где  $\delta_L$  - выбранная заранее высота расчётной области, превышающая толщину пограничного слоя на длине  $L$ ,  $\Delta x$  - шаг дискретизации по оси  $x$ ,  $\Delta y$  - шаг дискретизации по оси  $y$ . Дискретное множество точек и шаблон дискретизации представлены на рисунке ниже.



<sup>1</sup> Шлихтинг Г. Теория пограничного слоя. – М: Наука, 1974. – 713с.

Высоту расчётной области определим из оценок толщины пограничного слоя при ламинарном режиме течения:


$$\delta_L = p \cdot L \cdot \frac{5}{\sqrt{Re_L}} = p \cdot L \cdot \frac{5}{\sqrt{\frac{w_{x0} \cdot \rho \cdot L}{\mu}}}, \quad (3)$$

где:  $p$  - коэффициент запаса расчётной сетки.

На рисунке выше красным выделен шаблон дискретизации для уравнения движения при рассматриваемом **неявном методе решения уравнения**.

Следует помнить, что при выборе шаблона дискретизации для дифференциальных уравнений первого порядка необходимо предусмотреть минимум две точки по каждому из направлений, по которым берутся производные. Для уравнений второго порядка минимум три и т.д.

Для **неявного метода** решения уравнений пограничного слоя шаблоны дискретизации можно представить следующим образом:



Для уравнения движения – слева, для уравнения неразрывности - справа.

Получим выражения для производных, входящих в уравнения (1), в алгебраической форме, используя разложение функций в ряд Тейлора по соответствующей переменной, по аналогии с предыдущей лабораторной работой:

$$\left( \frac{\partial w_x}{\partial x} \right)_{i,j} = \frac{(w_x)_{i,j} - (w_x)_{i-1,j}}{\Delta x}, \quad \left( \frac{\partial w_y}{\partial y} \right)_{i,j} = \frac{(w_y)_{i,j} - (w_y)_{i,j-1}}{\Delta y} \quad \text{и} \quad \left( \frac{\partial w_x}{\partial y} \right)_{i,j} = \frac{(w_x)_{i,j} - (w_x)_{i,j-1}}{\Delta y}.$$

Для второй производной:

$$\left( \frac{\partial^2 w_x}{\partial y^2} \right)_{i,j} = \frac{(w_x)_{i,j+1} - 2 \cdot (w_x)_{i,j} + (w_x)_{i,j-1}}{\Delta y^2}.$$

Подставляя полученные соотношения в систему уравнений (1), можно записать дискретный аналог уравнений пограничного слоя:

$$\left\{ \begin{array}{l} (w_x)_{i,j} \left( \frac{(w_x)_{i,j} - (w_x)_{i-1,j}}{\Delta x} \right) + (w_y)_{i,j} \left( \frac{(w_x)_{i,j} - (w_x)_{i,j-1}}{\Delta y} \right) = v \left( \frac{(w_x)_{i,j+1} - 2(w_x)_{i,j} + (w_x)_{i,j-1}}{\Delta y^2} \right); \\ \frac{(w_x)_{i,j-1} - (w_x)_{i-1,j-1}}{\Delta x} + \frac{(w_y)_{i,j} - (w_y)_{i,j-1}}{\Delta y} = 0. \end{array} \right. \quad (4)$$

Используя **метод запаздывающих коэффициентов** для устранения нелинейности и сопряжённости уравнений (4) получим:

$$\left\{ \begin{aligned} (w_x)_{i-1,j} \left( \frac{(w_x)_{i,j} - (w_x)_{i-1,j}}{\Delta x} \right) + (w_y)_{i-1,j} \left( \frac{(w_x)_{i,j} - (w_x)_{i,j-1}}{\Delta y} \right) &= v \left( \frac{(w_x)_{i,j+1} - 2(w_x)_{i,j} + (w_x)_{i,j-1}}{\Delta y^2} \right); \\ \frac{(w_x)_{i,j-1} - (w_x)_{i-1,j-1}}{\Delta x} + \frac{(w_y)_{i,j} - (w_y)_{i,j-1}}{\Delta y} &= 0. \end{aligned} \right. \quad (5)$$

Следует обратить внимание на то, что на расчётном  $i$ -ом слое по координате  $x$  неизвестными в системе уравнений (5) являются сразу несколько переменных  $(w_x)_{i,j+1}, (w_x)_{i,j}, (w_x)_{i,j-1}$  и  $(w_y)_{i,j}$ . Однако каждая неизвестная связана с решением на соседних по  $j$  узлах расчётной сетки.

Преобразуем уравнение движения к виду:

$$\underbrace{\left( -\frac{(w_y)_{i-1,j}}{\Delta y} - \frac{v}{\Delta y^2} \right)}_{a_j} (w_x)_{i,j-1} + \underbrace{\left( \frac{(w_x)_{i-1,j}}{\Delta x} + \frac{(w_y)_{i-1,j}}{\Delta y} + 2\frac{v}{\Delta y^2} \right)}_{b_j} (w_x)_{i,j} + \underbrace{\left( -\frac{v}{\Delta y^2} \right)}_{c_j} (w_x)_{i,j+1} = \underbrace{\frac{(w_x)_{i-1,j}}{\Delta x} (w_x)_{i-1,j}}_{d_j}. \quad (6)$$

Обратим внимание на то, что коэффициенты при неизвестных значениях продольной скорости потока определены для каждой точки и не зависят от искомым переменных. Уравнение (6) является линейным уравнением типа  $ax_1 + bx_2 + cx_3 = d$ . Если записать уравнение (6) для каждого  $i, j$  узла, то можно получить систему линейных уравнений, где число неизвестных будет равно числу переменных. Разберём эту процедуру на примере расчётной сетки с 5 узлами (0,1,2,3,4) по высоте пограничного слоя, причём 0 будет соответствовать стенке, а 4 внешнему потоку:

$$\left\{ \begin{aligned} a_0 (w_x)_{i,-1} + b_0 (w_x)_{i,0} + c_0 (w_x)_{i,1} &= d_0 \\ a_1 (w_x)_{i,0} + b_1 (w_x)_{i,1} + c_1 (w_x)_{i,2} &= d_1 \\ a_2 (w_x)_{i,1} + b_2 (w_x)_{i,2} + c_2 (w_x)_{i,3} &= d_2 \\ a_3 (w_x)_{i,2} + b_3 (w_x)_{i,3} + c_2 (w_x)_{i,4} &= d_3 \\ a_4 (w_x)_{i,3} + b_4 (w_x)_{i,4} + \boxed{c_4 (w_x)_{i,5}} &= d_4 \end{aligned} \right. \quad (7)$$

Видно, что в уравнениях для 0 узла и для 4 узла присутствуют переменные, определённые вне расчётной области  $(w_x)_{i,-1}, (w_x)_{i,5}$ . Для исключения этих переменных воспользуемся граничными условиями, записанными в дискретной форме:  $(w_x)_{i,0} = 0$ ,  $(w_x)_{i,4} = w_{x0}$ :

$$\left\{ \begin{aligned} \boxed{0(w_x)_{i,-1}} + 1(w_x)_{i,0} + 0(w_x)_{i,1} = 0 &\Rightarrow \boxed{(w_x)_{i,0} = 0} \\ \boxed{0(w_x)_{i,3}} + 1(w_x)_{i,4} + \boxed{0(w_x)_{i,5}} = w_{x0} &\Rightarrow \boxed{(w_x)_{i,4} = w_{x0}} \end{aligned} \right. \quad (8)$$

Объединяя уравнения (7), (8) и записывая систему линейных уравнений в матричной форме, получим:

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ a_1 & b_1 & c_1 & 0 & 0 \\ 0 & a_2 & b_2 & c_2 & 0 \\ 0 & 0 & a_3 & b_3 & c_3 \\ 0 & 0 & 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} (w_x)_{i,0} \\ (w_x)_{i,1} \\ (w_x)_{i,2} \\ (w_x)_{i,3} \\ (w_x)_{i,4} \end{pmatrix} = \begin{pmatrix} 0 \\ d_1 \\ d_2 \\ d_3 \\ w_{x0} \end{pmatrix}. \quad (9)$$

Полученная матрица имеет только три диагонали с ненулевыми коэффициентами, называется **трёхдиагональной** и имеет эффективные методы решения. Одним из методов является **метод Томаса или метод простой прогонки**, которым мы и воспользуемся для решения системы уравнений (6).

Прежде чем перейти к описанию метода Томаса сделаем несколько замечаний.

Уравнение неразрывности в рассматриваемом методе решается также как и при явном методе решения уравнений пограничного слоя, рассмотренном на предыдущих лабораторных работах. Для решения уравнения неразрывности требуются граничные условия:  $(w_y)_{0,j} = 0$ ,  $(w_y)_{i,0} = 0$ .

Разработанная Нейманом линейная теория устойчивости не накладывает ограничений на устойчивость неявной схемы – схема считается **абсолютно устойчивой**. Для данной схемы шаг дискретизации по каждой из координат может быть выбран произвольно. Однако для решения системы уравнений (9) методом Томаса матрица коэффициентов должна обладать свойством **диагонального преобладания**:

$$b_j \geq a_j + c_j. \quad (10)$$

#### 4. МЕТОД ПРОГОНКИ (МЕТОД ТОМАСА)

Разберём простой однопоточный алгоритм решения трёхдиагональной матрицы методом прогонки подробно. Пусть имеем систему линейных уравнений пятого порядка:

$$\begin{aligned} b_1 X_1 + c_1 X_2 + 0X_3 + 0X_4 + 0X_5 &= d_1 \\ a_2 X_1 + b_2 X_2 + c_3 X_3 + 0X_4 + 0X_5 &= d_2 \\ 0X_1 + a_3 X_2 + b_3 X_3 + c_3 X_4 + 0X_5 &= d_3 \\ 0X_1 + 0X_2 + a_4 X_3 + b_4 X_4 + c_4 X_5 &= d_4 \\ 0X_1 + 0X_2 + 0X_3 + a_5 X_4 + b_5 X_5 &= d_5 \end{aligned} \quad (11)$$

В матричном виде можно записать:

$$\begin{pmatrix} b_1 & c_1 & 0 & 0 & 0 \\ a_2 & b_2 & c_2 & 0 & 0 \\ 0 & a_3 & b_3 & c_3 & 0 \\ 0 & 0 & a_4 & b_4 & c_4 \\ 0 & 0 & 0 & a_5 & b_5 \end{pmatrix} \begin{pmatrix} X_1 \\ X_2 \\ X_3 \\ X_4 \\ X_5 \end{pmatrix} = \begin{pmatrix} d_1 \\ d_2 \\ d_3 \\ d_4 \\ d_5 \end{pmatrix} \quad (12)$$

##### Прямая прогонка:

Смысл прямой прогонки заключается в исключении нижней диагонали матрицы (или обнулении коэффициентов  $a_i$ ). Первое уравнение системы (11) остаётся без изменения ( $a_1 = 0$ ), но из него получаем соотношения:

$$b_1 X_1 + c_1 X_2 = d_1 \quad (13)$$

$$X_1 = \frac{d_1}{b_1} - \frac{c_1}{b_1} X_2 \quad (14)$$

Второе уравнение системы (11) с учётом (14) записывается следующим образом:

$$\underbrace{\left( b_2 - \frac{a_2 c_1}{b_1} \right)}_{B_2} X_2 + c_2 X_3 = \underbrace{d_2 - \frac{a_2 d_1}{b_1}}_{D_2} \quad (15)$$

Из (15) получаем:

$$X_2 = \frac{\left( d_2 - \frac{a_2 d_1}{b_1} \right)}{\left( b_2 - \frac{a_2 c_1}{b_1} \right)} - \frac{c_2}{\left( b_2 - \frac{a_2 c_1}{b_1} \right)} X_3 \quad (16)$$

Далее по аналогии:

$$\underbrace{\left( b_3 - \frac{a_3 c_2}{\left( b_2 - \frac{a_2 c_1}{b_1} \right)} \right)}_{B_3} X_3 + c_3 X_4 = \underbrace{\left( d_3 - a_3 \frac{\left( d_2 - \frac{a_2 d_1}{b_1} \right)}{\left( b_2 - \frac{a_2 c_1}{b_1} \right)} \right)}_{D_3} \quad (17)$$

$$X_3 = \frac{\left( d_3 - a_3 \frac{\left( d_2 - \frac{a_2 d_1}{b_1} \right)}{\left( b_2 - \frac{a_2 c_1}{b_1} \right)} \right)}{\left( b_3 - \frac{a_3 c_2}{\left( b_2 - \frac{a_2 c_1}{b_1} \right)} \right)} - \frac{c_3}{\left( b_3 - \frac{a_3 c_2}{\left( b_2 - \frac{a_2 c_1}{b_1} \right)} \right)} X_4 \quad (18)$$

$$\underbrace{\left( b_4 - \frac{a_4 c_3}{\left( b_3 - \frac{a_3 c_2}{\left( b_2 - \frac{a_2 c_1}{b_1} \right)} \right)} \right)}_{B_4} X_4 + c_4 X_5 = \underbrace{\left( d_4 - a_4 \frac{\left( d_3 - a_3 \frac{\left( d_2 - \frac{a_2 d_1}{b_1} \right)}{\left( b_2 - \frac{a_2 c_1}{b_1} \right)} \right)}{\left( b_3 - \frac{a_3 c_2}{\left( b_2 - \frac{a_2 c_1}{b_1} \right)} \right)} \right)}_{D_4} \quad (19)$$

$$X_4 = \frac{\left( \left( \left( \left( \left( d_3 - a_3 \left( \frac{d_2 - \frac{a_2 d_1}{b_1}}{b_2 - \frac{a_2 c_1}{b_1}} \right) \right) \right) \right) \right) \right) \left( b_3 - \frac{a_3 c_2}{b_2 - \frac{a_2 c_1}{b_1}} \right)}{d_4 - a_4} - \frac{c_4}{X_5} \cdot \left( b_4 - \frac{a_4 c_3}{b_3 - \frac{a_3 c_2}{b_2 - \frac{a_2 c_1}{b_1}}} \right) \tag{20}$$

$$X_5 = d_5 - a_5 \cdot \left( b_5 - \frac{a_5 c_4}{b_4 - \frac{a_4 c_3}{b_3 - \frac{a_3 c_2}{b_2 - \frac{a_2 c_1}{b_1}}} \right) \cdot \left( \left( \left( \left( \left( \left( d_4 - a_4 \left( \frac{d_3 - a_3 \left( \frac{d_2 - \frac{a_2 d_1}{b_1}}{b_2 - \frac{a_2 c_1}{b_1}} \right) \right) \right) \right) \right) \right) \right) \left( b_3 - \frac{a_3 c_2}{b_2 - \frac{a_2 c_1}{b_1}} \right) \right) \tag{21}$$

Используя соотношения (13),(15),(17),(19),(21), получаем матрицу коэффициентов в следующем виде:

$$\begin{pmatrix} b_1 & c_1 & 0 & 0 & 0 \\ 0 & B_2 & c_2 & 0 & 0 \\ 0 & 0 & B_3 & c_3 & 0 \\ 0 & 0 & 0 & B_4 & c_4 \\ 0 & 0 & 0 & 0 & B_5 \end{pmatrix} \begin{pmatrix} X_1 \\ X_2 \\ X_3 \\ X_4 \\ X_5 \end{pmatrix} = \begin{pmatrix} d_1 \\ D_2 \\ D_3 \\ D_4 \\ D_5 \end{pmatrix}, \quad (22)$$

где, как можно заметить:

$$\begin{aligned} B_5 &= b_5 - \frac{a_5 c_4}{B_4} & D_5 &= d_5 - a_5 \frac{D_4}{B_4} \\ B_4 &= b_4 - \frac{a_4 c_3}{B_3} & D_4 &= d_4 - a_4 \frac{D_3}{B_3} \\ B_3 &= b_3 - \frac{a_3 c_2}{B_2} & D_3 &= d_3 - a_3 \frac{D_2}{B_2}, \quad i = 2 \dots n \\ B_2 &= b_2 - \frac{a_2 c_1}{b_1} & D_2 &= d_2 - a_2 \frac{d_1}{b_1} \end{aligned} \quad (23)$$

$$B_i = b_i - \frac{a_i c_{i-1}}{B_{i-1}} \quad D_i = d_i - a_i \frac{D_{i-1}}{B_{i-1}}$$

### Обратная прогонка:

Смысл обратной прогонки заключается в исключении верхней диагонали матрицы (22) и преобразовании полученной матрицы к единичной:

Из пятого уравнения системы (22) получаем

$$X_5 = \frac{D_5}{B_5} = \Delta_5 \quad (24)$$

Из четвертого уравнения системы (22) с учётом (23):

$$B_4 X_4 + c_4 \frac{D_5}{B_5} = D_4 \quad (25)$$

$$X_4 = \frac{D_4 - c_4 \frac{D_5}{B_5}}{B_4} \quad (26)$$

Далее по аналогии:

$$B_3 X_3 + c_3 \frac{D_4 - c_4 \frac{D_5}{B_5}}{B_4} = D_3 \quad (27)$$

$$X_3 = \frac{D_3 - c_3 \frac{D_4 - c_4 \frac{D_5}{B_5}}{B_4}}{B_3} = \Delta_3 \quad (28)$$



$$B_2 X_2 + c_2 \frac{D_3 - c_3 \frac{D_4 - c_4 \frac{D_5}{B_5}}{B_4}}{B_3} = D_2 \quad (29)$$

$$X_2 = \frac{D_2 - c_2 \frac{D_3 - c_3 \frac{D_4 - c_4 \frac{D_5}{B_5}}{B_4}}{B_3}}{B_2} = \Delta_2 \quad (30)$$

$$b_1 X_1 + c_1 \frac{D_2 - c_2 \frac{D_3 - c_3 \frac{D_4 - c_4 \frac{D_5}{B_5}}{B_4}}{B_3}}{B_2} = d_1 \quad (31)$$

$$X_1 = \frac{d_1 - c_1 \frac{D_2 - c_2 \frac{D_3 - c_3 \frac{D_4 - c_4 \frac{D_5}{B_5}}{B_4}}{B_3}}{B_2}}{b_1} = \Delta_1 \quad (32)$$

Используя соотношения (24),(26),(28),(30),(32) получаем матрицу коэффициентов в следующем виде:

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} X_1 \\ X_2 \\ X_3 \\ X_4 \\ X_5 \end{pmatrix} = \begin{pmatrix} \Delta_1 \\ \Delta_2 \\ \Delta_3 \\ \Delta_4 \\ \Delta_5 \end{pmatrix}, \quad (33)$$

где, как можно заметить:

$$\begin{aligned} \Delta_5 &= \frac{D_5}{B_5} \\ \Delta_4 &= \frac{D_4 - c_4 \Delta_5}{B_4} \\ \Delta_3 &= \frac{D_3 - c_3 \Delta_4}{B_3} \\ \Delta_2 &= \frac{D_2 - c_2 \Delta_3}{B_2} \\ \Delta_1 &= \frac{d_1 - c_1 \Delta_2}{b_1} \end{aligned} \quad i = n-1 \dots 1 \quad (34)$$

$$\boxed{\Delta_i = \frac{D_i - c_i \Delta_{i+1}}{B_i}}$$

Обобщённо алгоритм Томаса можно представить в следующем виде:

$$\begin{pmatrix} b_1 & c_1 & 0 & & \dots & 0 \\ a_2 & b_2 & c_2 & 0 & & 0 \\ 0 & a_3 & b_3 & c_3 & 0 & \dots \\ \dots & & & & & \\ 0 & \dots & 0 & a_i & b_i & c_i \\ \dots & & & & & \\ 0 & & \dots & 0 & a_n & b_n \end{pmatrix} \begin{pmatrix} X_1 \\ X_2 \\ X_3 \\ \dots \\ X_i \\ \dots \\ X_n \end{pmatrix} = \begin{pmatrix} d_1 \\ d_2 \\ d_3 \\ \dots \\ d_i \\ \dots \\ d_n \end{pmatrix} \xrightarrow{\substack{i=2\dots n \\ B_i = b_i - \frac{a_i c_{i-1}}{B_{i-1}} \\ D_i = d_i - \frac{a_i D_{i-1}}{B_{i-1}}} }$$

$$\begin{pmatrix} b_1 & c_1 & 0 & & \dots & 0 \\ 0 & B_2 & c_2 & 0 & & 0 \\ 0 & 0 & B_3 & c_3 & 0 & \dots \\ \dots & & & & & \\ 0 & \dots & 0 & 0 & B_i & c_i \\ \dots & & & & & \\ 0 & & \dots & 0 & 0 & B_n \end{pmatrix} \begin{pmatrix} X_1 \\ X_2 \\ X_3 \\ \dots \\ X_i \\ \dots \\ X_n \end{pmatrix} = \begin{pmatrix} D_1 \\ D_2 \\ D_3 \\ \dots \\ D_i \\ \dots \\ D_n \end{pmatrix} \xrightarrow{\substack{i=n-1\dots 1 \\ \Delta_n = \frac{D_n}{B_n} \\ \Delta_i = \frac{D_i - c_i \Delta_{i+1}}{B_i}}}$$

$$\begin{pmatrix} 1 & 0 & 0 & & \dots & 0 \\ 0 & 1 & 0 & 0 & & 0 \\ 0 & 0 & 1 & 0 & 0 & \dots \\ \dots & & & & & \\ 0 & \dots & 0 & 0 & 1 & 0 \\ \dots & & & & & \\ 0 & & \dots & 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} X_1 \\ X_2 \\ X_3 \\ \dots \\ X_i \\ \dots \\ X_n \end{pmatrix} = \begin{pmatrix} \Delta_1 \\ \Delta_2 \\ \Delta_3 \\ \dots \\ \Delta_i \\ \dots \\ \Delta_n \end{pmatrix}$$

Реализация алгоритма на языке программирования Fortran90, с использованием модулей и пользовательских типов данных:

### Модуль mathutilits:

```

module mathutilits
implicit none

! Тип данных о трёхдиагональной матрице
type tdma_rec
integer*4      :: b_point
integer*4      :: n_point
real*8, allocatable :: a(:)
real*8, allocatable :: b(:)
real*8, allocatable :: c(:)
real*8, allocatable :: d(:)
end type tdma_rec

! Трёхдиагональная матрица
type(tdma_rec) :: tdma

contains

!-----
! Подпрограммы решения трёхдиагональной матрицы

```

```

! Инициализация трёхдиагональной матрицы
subroutine tdma_init (tdmav)
type (tdma_rec), intent (inout) :: tdmav
  allocate (tdmav%a (tdmav%b_point:tdmav%n_point))
  allocate (tdmav%b (tdmav%b_point:tdmav%n_point))
  allocate (tdmav%c (tdmav%b_point:tdmav%n_point))
  allocate (tdmav%d (tdmav%b_point:tdmav%n_point))
  tdmav%a=0d0
  tdmav%b=0d0
  tdmav%c=0d0
  tdmav%d=0d0
end subroutine tdma_init

! Деинициализация трёхдиагональной матрицы
subroutine tdma_deinit (tdmav)
type (tdma_rec), intent (inout) :: tdmav
  deallocate (tdmav%a,tdmav%b,tdmav%c,tdmav%d)
end subroutine tdma_deinit

! Расчёт трёхдиагональной матрицы методом Томаса (методом прямой прогонки)
subroutine tdma_progonka (tdmav)
type (tdma_rec), intent (inout) :: tdmav
integer*4 :: i,j
  do i=tdmav%b_point+1,tdmav%n_point
    tdmav%b (i)=tdmav%b (i) - (tdmav%a (i) *tdmav%c (i-1) /tdmav%b (i-1))
    tdmav%d (i)=tdmav%d (i) - (tdmav%a (i) *tdmav%d (i-1) /tdmav%b (i-1))
    tdmav%a (i)=0d0
  end do
  tdmav%d (tdmav%n_point)=tdmav%d (tdmav%n_point) /tdmav%b (tdmav%n_point)
  tdmav%b (tdmav%n_point)=1d0
  do j=tdmav%n_point-1,tdmav%b_point,-1
    tdmav%d (j)=(tdmav%d (j) -tdmav%c (j) *tdmav%d (j+1)) /tdmav%b (j)
    tdmav%c (j)=0d0
    tdmav%b (j)=1d0
  end do
end subroutine tdma_progonka

end module mathutilits

```

## Применение модуля mathutilits и результаты расчёта:

```

program tomas
use mathutilits
implicit none

tdma%b_point = 1
tdma%n_point = 10
call tdma_init (tdma)

tdma%a = 0.3d0
tdma%b = 0.2d0
tdma%c = 0.5d0
tdma%d = 0.8d0

call tdma_progonka (tdma)

call tdma_deinit (tdma)
end program tomas

```

## 5. ПОРЯДОК ПРОВЕДЕНИЯ МОДЕЛИРОВАНИЯ

1. Получить задание для моделирования течения в ламинарном пограничном слое на плоской пластине:

Среда:

Скорость набегающего потока:

Температура набегающего потока:

Длина пластины:

2. Подготовить программу на языке Fortran, реализующую метод решения трёхдиагональной матрицы методом Томаса.

3. Провести компиляцию и отладку программы. Протестировать разработанную программу на приведённом примере расчёта трёхдиагональной матрицы размером 10x10.

Исходные данные:

	a	b	c	d
1	.0000	.2000	.5000	.8000
2	.3000	.2000	.5000	.8000
3	.3000	.2000	.5000	.8000
4	.3000	.2000	.5000	.8000
5	.3000	.2000	.5000	.8000
6	.3000	.2000	.5000	.8000
7	.3000	.2000	.5000	.8000
8	.3000	.2000	.5000	.8000
9	.3000	.2000	.5000	.8000
10	.3000	.2000	.0000	.8000

Полученный результат:

	a	b	c	d
1	.0000	1.0000	.0000	-9.1846
2	.0000	1.0000	.0000	5.2739
3	.0000	1.0000	.0000	5.0012
4	.0000	1.0000	.0000	-3.5648
5	.0000	1.0000	.0000	.0252
6	.0000	1.0000	.0000	3.7288
7	.0000	1.0000	.0000	.0934
8	.0000	1.0000	.0000	-.6746
9	.0000	1.0000	.0000	1.8138
10	.0000	1.0000	.0000	1.2792

4. Подготовить программу на языке Fortran, реализующую неявный метод численного интегрирования уравнений динамического пограничного слоя с устранением нелинейности методом запаздывающих коэффициентов.

5. Провести компиляцию и отладку программы.

6. Выполнить расчёты профилей скорости на расстоянии 20, 40, 60, 80, 100 % от длины пластины. Построить графики в программе Excel, Origin или др. Сделать выводы о росте толщины пограничного слоя. Занести полученные данные в отчёт.

6. Выполнить расчёты интегральных параметров течения (локального коэффициента поверхностного трения) в зависимости от локального числа Рейнольдса для различного кол-ва узлов расчётной сетки по осям x и y. Построить график зависимости коэффициента трения от числа Рейнольдса в логарифмических координатах. Нанести на график формулу Блазиуса. Сделать выводы о влиянии сетки на результаты расчётов интегральных параметров течения. Занести полученные результаты в отчёт.

7. Занести текст программы в отчёт.

8. Оформить отчёт. Подготовиться к защите лабораторной работы.

## **6. СОДЕРЖАНИЕ ОТЧЁТА**

1. Титульный лист.
2. Цель работы.
3. Схема решаемой задачи с указанием расчётной области, граничных условий и параметров задачи.
4. Основные формулы, используемые при расчетах.
5. Результаты моделирования.
6. Тексты программ.
7. Выводы по работе.

## **7. КОНТРОЛЬНЫЕ ВОПРОСЫ**

1. Что называется пограничным слоем?
2. На каком шаблоне записывается дискретная форма уравнений пограничного слоя при неявном методе решения?
3. Как влияет шаг дискретизации в направлении поперёк течения на точность определения коэффициента трения в сравнении с формулой Блазиуса?
4. Критерии устойчивости неявного метода?
5. Что такое трёхдиагональная матрица?
6. Что такое диагональное преобладание?
7. Как записываются граничные условия в дискретной форме при решении уравнения неявным методом?